

MyTitle

MyName

November 4, 2024

# Contents

<b>1</b>	<b>Introduction</b>	<b>2</b>
<b>2</b>	<b>Preliminaries</b>	<b>4</b>
2.1	Set-Theoretic Preliminaries . . . . .	4
2.1.1	ZF Set Theory and the Axiom of Choice . . . . .	4
2.1.2	Forcing . . . . .	5
2.1.3	Symmetric Extensions . . . . .	7
2.1.4	The c.t.m. Approach . . . . .	8
2.2	Outline for the Informal Proof . . . . .	8
2.3	Isabelle/ZF and Formalization in Prior Work . . . . .	9
<b>3</b>	<b>Formalization of the Proof</b>	<b>10</b>
3.1	Defining Symmetric Extensions . . . . .	10
3.2	Proving Symmetric Extensions are Models of ZF . . . . .	14
3.3	Constructing a Model of $ZF + \neg AC$ . . . . .	17
<b>4</b>	<b>Conclusion and Future Work</b>	<b>21</b>

# Chapter 1

## Introduction

The formalization of mathematics using proof assistants such as Isabelle, Coq, and Lean, has been actively conducted, leading to numerous achievements. For instance, the proofs of the four color theorem, Kepler's conjecture, and the Feit-Thompson theorem have been formalized using proof assistants, enhancing the reliability of these proofs. Additionally, various fields of mathematics, such as number theory, algebra, and topology, are also being formalized.

The independence of the axiom of choice (AC) from Zermelo-Fraenkel set theory (ZF) is a well-known result in the early history of axiomatic set theory, as well as the independence of the continuum hypothesis (CH) from ZF with AC (ZFC). Cohen invented the forcing method and proved them in 1963. Forcing is a powerful tool for exploring models of set theory and was subsequently further sophisticated by other researchers.

Independence proofs of CH from ZFC has been formalized in Isabelle/ZF by Gunther et al. [2] and in Lean 3 by Han and van Doorn [3]. In these studies, forcing methods were formalized, and the independence of CH was proven by showing the relative consistency of CH and  $\neg$ CH with ZFC.

For AC, the relative consistency of AC with ZF has been formalized in Isabelle/ZF by Paulson [11]. However, the relative consistency of  $\neg$ AC with ZF has not been formalized. It can be proven by forcing, but the proof involves complexities that cannot be achieved by simply modifying the proof for CH.

In this work, we formalized the relative consistency proof of  $\neg$ AC with ZF in Isabelle/ZF. This work contributes to the formalization of axiomatic set theory and serves as a new example of the formalization using forcing, which is a crucial tool in set theory. It also may provide insights into how the formalization of axiomatic set theory could be advanced.

### Our Approach

The primary reason we chose Isabelle/ZF for this formalization is that Isabelle/ZF is a mature proof assistant for ZF set theory, in particular, the formalization by Gunther et al. [1] is a major advantage for this study. Although there is also a formalization of forcing in Lean 3 by Han and van Doorn [3], development for Lean 3 has already ended.

To use Gunther et al.’s formalization, we adopted the c.t.m. approach for our proof, as in their independence proof of CH. Specifically, we assumed the existence of a c.t.m. of ZF and constructed a model of  $ZF + \neg AC$  by forcing. This model is known as the basic Cohen model, which is a symmetric extension of assumed c.t.m. Our proof is based on Karagila’s lecture note [8] and Jech’s book [5, 6], as these resources align well with this approach.

The c.t.m. approach is a well-established method for relative consistency proofs in axiomatic set theory. The relative consistency of  $\neg AC$  with ZF means that if ZF is consistent, then  $ZF + \neg AC$  is also consistent. Assuming the consistency of ZF implies that a model of ZF exists, but strictly speaking, the existence of a c.t.m. cannot be derived from this assumption in ZF. However, as explained in section 2.1.4, we can prove the relative consistency of  $\neg AC$  with ZF if we can construct a model of  $ZF + \neg AC$  from a c.t.m. of ZF. This kind of reasoning always arises in the c.t.m. approach, and ideally, we would like to formalize it as well, but this has not been achieved.

### **Related works**

Paulson et al. formalized an extensive part of ZF set theory [9, 10, 11, 12, 13], including cardinal arithmetic, relativization, the reflection theorem, features for handling inductive definitions, and the relative consistency of AC with ZF. The proof of the relative consistency was achieved by constructing Gödel’s constructible universe.

Building on these results, Gunther et al. formalized forcing and a proof of the independence of CH [1, 2] in Isabelle/ZF. In these formalizations, the countable transitive model (c.t.m.) approach was used, following Kunen’s book [7].

In Lean, Han and van Doorn also formalized forcing and the independence of CH [3] in Lean 3. using the Boolean-valued model approach, which is another approach to forcing. Additionally, in Lean 4, Holmes and Wilshaw formalized the complex parts of the consistency proof of Quine’s New Foundations [4], ensuring the correctness of the proof.

### **Repository**

Our source code is available at: [https://github.com/tarakojo/ZF\\_notAC](https://github.com/tarakojo/ZF_notAC)

## Chapter 2

# Preliminaries

### 2.1 Set-Theoretic Preliminaries

In this section, we introduce the concepts of set theory used in the formalization of this study. Our meta-theory is ZF, within which we explore ZF itself. Basically, our definition and theorems combine Kunen[7], karagila's lecture note[8], and Jech's book[6] adapted to the formalized form in Isabelle/ZF.

#### 2.1.1 ZF Set Theory and the Axiom of Choice

We introduce the axioms of ZF set theory and the axiom of choice(AC). We use first-order logic with the language of set theory, which consists only of only two relation symbols  $\in$  and  $=$ . Formulas involving other mathematical operators that may appear are considered abbreviations for formulas in this language. Parentheses in formulas are omitted where no confusion arises.

**Definition 2.1.1.** *The axioms of ZF are the following statements:*

- *Extensionality:*  $\forall x \forall y (\forall z (z \in x \leftrightarrow z \in y) \rightarrow x = y)$
- *Pairing:*  $\forall x \forall y \exists z \forall w (w \in z \leftrightarrow w = x \vee w = y)$
- *Union:*  $\forall x \exists y \forall z (z \in y \leftrightarrow \exists w (z \in w \wedge w \in x))$
- *Power set:*  $\forall x \exists y \forall z (z \in y \leftrightarrow z \subseteq x)$
- *Infinity:*  $\exists x (\emptyset \in x \wedge \forall y (y \in x \rightarrow y \cup \{y\} \in x))$
- *Regularity:*  $\forall x (x \neq \emptyset \rightarrow \exists y (y \in x \wedge y \cap x = \emptyset))$
- *Infinity:*  $\forall x (x \neq \emptyset \rightarrow \exists y (y \in x \wedge \forall z (z \in x \rightarrow z \notin y)))$
- *Separation:*  $\forall p \forall x \exists y \forall z (z \in y \leftrightarrow z \in x \wedge \phi(z, p))$
- *Replacement:*  $\forall p (\forall x \forall y \forall z (\phi(x, y, p) \wedge \phi(x, z, p) \rightarrow y = z) \rightarrow \forall X \exists Y \forall y (y \in Y \leftrightarrow \exists x (x \in X \wedge \phi(x, y, p))))$

Where separation and replacement are axiom schemas, representing infinitely many axioms for each formula  $\phi$  with an appropriate arity.

**Definition 2.1.2.** The axiom of choice (AC) is the following statement:  
 $\forall x \exists f ("f \text{ is a function on } x" \wedge \forall y (y \in x \rightarrow f(y) \in y))$

Where the phrase "f is a function on x" is also considered an abbreviation in the language of set theory. Theory ZF + AC is denoted by ZFC. Additionally, we introduce the well-ordering theorem, as we treat AC in this form.

**Definition 2.1.3.** We say that a linear ordering  $<$  on a set  $P$  is a well-ordering if, every non-empty subset of  $P$ , it has a least element.

**Lemma 2.1.4.** The axiom of choice is equivalent to the well-ordering theorem, which states that every set can be well-ordered.

### 2.1.2 Forcing

Forcing is a technique used in proving relative consistency and Independence. We introduce basic concepts of forcing in the context of the c.t.m. approach. In this approach, the relative consistency proof is achieved by using forcing to construct an extended model by adding new sets to an assumed c.t.m. Let  $M$  be a c.t.m. of ZF and  $(\mathbb{P}, \leq_{\mathbb{P}})$  be a notion of forcing, which is a pre-ordered set in  $M$  with a maximum element  $1_{\mathbb{P}}$ .

**Definition 2.1.5.** We define  $M^{\mathbb{P}}$ , the set of  $\mathbb{P}$ -names, by transfinite recursion on ordinals:

1.  $M_0^{\mathbb{P}} = \emptyset$
2.  $M_{\alpha+1}^{\mathbb{P}} = \mathcal{P}^M(M_{\alpha}^{\mathbb{P}} \times \mathbb{P})$
3.  $M_{\alpha}^{\mathbb{P}} = \bigcup_{\beta < \alpha} M_{\beta}^{\mathbb{P}}$  for a limit ordinal  $\alpha$
4.  $M^{\mathbb{P}} = \bigcup_{\alpha \in \text{Ord}} M_{\alpha}^{\mathbb{P}}$

Where  $\mathcal{P}^M$  denotes the power set operation in  $M$ . We often write a  $\mathbb{P}$ -name with a dot, e.g.,  $\dot{x}$ . The least  $\alpha$  such that  $\dot{x} \in M_{\alpha+1}^{\mathbb{P}}$  is called the  $\mathbb{P}$ -rank of  $\dot{x}$ . This allows us to define functions and relations by recursion on  $\mathbb{P}$ -names.

**Definition 2.1.6.**

1. We say that  $D \subseteq \mathbb{P}$  is dense if, for every  $p \in \mathbb{P}$ , there exists  $q \in D$  such that  $q \leq_{\mathbb{P}} p$ .
2. We say that  $G \subseteq \mathbb{P}$  is a filter if following conditions hold:
  - If  $p \in G$ ,  $q \in \mathbb{P}$ , and  $p \leq_{\mathbb{P}} q$ , then  $q \in G$
  - If  $p, q \in G$ , there exists  $r \in G$  such that  $r \leq_{\mathbb{P}} p$  and  $r \leq_{\mathbb{P}} q$
3. We say that  $G \subseteq \mathbb{P}$  is generic filter on  $\mathbb{P}$  if  $G$  is a filter and for any dense  $D \subseteq \mathbb{P}$ ,  $D \cap G \neq \emptyset$ .

The following lemma shows that a generic filter actually exists.

**Lemma 2.1.7.** *For any  $p \in \mathbb{P}$ , there exists a generic filter  $G$  on  $\mathbb{P}$  such that  $p \in G$ .*

**Definition 2.1.8.** *Let  $G$  be a generic filter on  $\mathbb{P}$  and  $\dot{x} \in M^{\mathbb{P}}$ . We define the interpretation of  $\dot{x}$  denoted by  $\dot{x}^G$  recursively with respect to the  $\mathbb{P}$ -rank of  $\dot{x}$ :*

$$\dot{x}^G = \{\dot{y}^G \mid \exists p \in G (\langle \dot{y}, p \rangle \in \dot{x})\}$$

We call a  $\mathbb{P}$ -name whose interpretation is a set  $x$  a name of  $x$  and denote it by  $\dot{x}$ . Note that a single set may have multiple names.

**Definition 2.1.9.** *Let  $G$  be a generic filter on  $\mathbb{P}$ . We define a generic extension  $M[G]$  as  $\{\dot{x}^G \mid \dot{x} \in M^{\mathbb{P}}\}$ .*

**Theorem 2.1.10.** *Let  $G$  be a generic filter on  $\mathbb{P}$ . Then,  $M[G]$  is the smallest c.t.m. of ZF extending  $M$  and containing  $G$ .*

By choosing  $\mathbb{P}$  appropriately, we can construct  $M[G]$  with various properties. What holds or does not hold in  $M[G]$  can be identified using the forcing relation. The forcing relation is defined recursively on formulas in a forcing language. The forcing language is an extension of the language of set theory by adding the elements of  $M^{\mathbb{P}}$  as constants.

**Definition 2.1.11.** *We define the forcing relation  $\Vdash$  for formulas in the forcing language and  $p \in \mathbb{P}$  inductively<sup>1</sup> as follows:*

1.  $p \Vdash \dot{x} = \dot{y} \Leftrightarrow \forall \dot{z} \in \text{dom}(\dot{x}) \cup \text{dom}(\dot{y}) \forall q \leq_{\mathbb{P}} p (q \Vdash \dot{z} \in \dot{x} \leftrightarrow q \Vdash \dot{z} \in \dot{y})$
2.  $p \Vdash \dot{x} \in \dot{y} \Leftrightarrow \forall q \leq_{\mathbb{P}} p \exists r \leq_{\mathbb{P}} q \exists s \in \mathbb{P} \exists \dot{z} \in M^{\mathbb{P}} (\langle \dot{z}, s \rangle \in \dot{y} \wedge r \leq_{\mathbb{P}} s \wedge r \Vdash \dot{x} = \dot{z})$
3.  $p \Vdash \phi \wedge \psi \Leftrightarrow p \Vdash \phi \wedge p \Vdash \psi$
4.  $p \Vdash \neg \phi \Leftrightarrow \forall q \leq_{\mathbb{P}} p \neg (q \Vdash \phi)$
5.  $p \Vdash \exists x \phi(x) \Leftrightarrow \forall q \leq_{\mathbb{P}} p \exists r \leq_{\mathbb{P}} q \exists \dot{x} \in M^{\mathbb{P}} (r \Vdash \phi(\dot{x}))$

**Theorem 2.1.12** (The Truth Lemma). *Let  $G$  be a generic filter on  $\mathbb{P}$ ,  $\varphi$  be a formula, and  $\dot{x} \in M^{\mathbb{P}}$ , then*

$$M[G] \models \varphi(\dot{x}^G) \Leftrightarrow \exists p \in G (p \Vdash \varphi(\dot{x}))$$

**Corollary 2.1.13.** *Let  $p \in \mathbb{P}$ ,  $\varphi$  be a formula and  $\dot{x} \in M^{\mathbb{P}}$ , then*

$$p \Vdash \varphi(\dot{x}) \Leftrightarrow \text{for any generic filter } G \text{ containing } p, M[G] \models \varphi(\dot{x}^G)$$

---

<sup>1</sup>Specifically, the forcing relation is first defined for atomic formulas  $\dot{x} = \dot{y}$  and  $\dot{x} \in \dot{y}$  by mutual recursion, inductively on the  $\mathbb{P}$ -rank of  $\dot{x}$  and  $\dot{y}$ . Then, the definition is extended to all formulas in the forcing language by induction on the complexity of formulas.

### 2.1.3 Symmetric Extensions

Symmetric extensions are substructures of generic extensions of a given c.t.m. of ZF and are formed by interpreting only the hereditarily symmetric names. Let  $M$  be a c.t.m. of ZF,  $(\mathbb{P}, \leq_{\mathbb{P}})$  be a pre-ordered set in  $M$  with the maximum element  $1_{\mathbb{P}}$ .

**Definition 2.1.14.** We say that  $\pi : \mathbb{P} \rightarrow \mathbb{P}$  is an automorphism if for all  $p, q \in \mathbb{P}$ ,  $p \leq_{\mathbb{P}} q \Leftrightarrow \pi p \leq_{\mathbb{P}} \pi q$ .  $\pi$  induces an bijection on  $\mathbb{P}$ -names defined recursively as follows:

$$\pi \dot{x} = \{ \langle \pi \dot{y}, \pi p \rangle \mid \langle \dot{y}, p \rangle \in \dot{x} \}$$

**Definition 2.1.15.** Let  $\mathcal{G}$  be a group of automorphisms of  $\mathbb{P}$ . We say that  $\mathcal{F}$  is a normal filter on  $\mathcal{G}$  if the following conditions hold:

1.  $\mathcal{F}$  is non-empty family of subgroups of  $\mathcal{G}$ .
2.  $\mathcal{F}$  is closed under finite intersections and supergroups.
3. For every  $H \in \mathcal{F}$  and  $\pi \in \mathcal{G}$ ,  $\pi H \pi^{-1} \in \mathcal{F}$ .

We fix a group of automorphisms  $\mathcal{G}$  of  $\mathbb{P}$  and a normal filter  $\mathcal{F}$  on  $\mathcal{G}$ .

**Definition 2.1.16.** For every  $\mathbb{P}$ -name  $\dot{x}$ , let  $\text{sym}(\dot{x}) = \{ \pi \in \mathcal{G} \mid \pi \dot{x} = \dot{x} \}$ . We say that  $\mathbb{P}$ -name  $\dot{x}$  is symmetric if  $\text{sym}(\dot{x}) \in \mathcal{F}$ .

**Definition 2.1.17.** We define the set of all hereditarily symmetric names  $\text{HS}$  recursively as follows:

1.  $0 \in \text{HS}$
2. if  $\text{dom}(\dot{x}) \subseteq \text{HS}$  and  $\dot{x}$  is symmetric, then  $\dot{x} \in \text{HS}$

**Definition 2.1.18.** Let  $G$  be a generic filter on  $\mathbb{P}$ . The set  $\text{HS}^G = \{ \dot{x}^G \mid \dot{x} \in \text{HS} \}$  is called a symmetric extension of  $M$ .

**Theorem 2.1.19.** Let  $G$  be a generic filter on  $\mathbb{P}$ . Then, the symmetric extension  $\text{HS}^G$  is a c.t.m. of ZF and a substructure of  $M[G]$ .

**Definition 2.1.20.** The relativized forcing relation  $\Vdash_{\text{HS}}$  is defined as the forcing relation  $\Vdash$  with  $M^{\mathbb{P}}$  in its definition replaced by  $\text{HS}$ .

The relation  $\Vdash_{\text{HS}}$  acts as the forcing relation for symmetric extensions.

**Theorem 2.1.21.** Let  $G$  be a generic filter on  $\mathbb{P}$ ,  $\mathcal{N}$  be a symmetric extension generated by  $G$ ,  $\varphi$  be a formula, and  $\dot{x} \in \text{HS}$ , then

$$\mathcal{N} \models \varphi(\dot{x}) \Leftrightarrow \exists p \in G (p \Vdash_{\text{HS}} \varphi(\dot{x}))$$

**Corollary 2.1.22.** Let  $p \in \mathbb{P}$ ,  $\varphi$  be a formula, and  $\dot{x} \in \text{HS}$ , then

$$p \Vdash_{\text{HS}} \varphi(\dot{x}) \Leftrightarrow \text{for any generic filter } G \text{ containing } p, \mathcal{N} \models \varphi(\dot{x}^G)$$

Where  $\mathcal{N}$  is the symmetric extension generated by  $G$ .



The following lemmas also holds for the forcing relation  $\Vdash$ , but we only state them for  $\Vdash_{\text{HS}}$ .

**Lemma 2.1.23.** *Let  $p, q \in \mathbb{P}$ ,  $\varphi$  be a formula. If  $q \leq_{\mathbb{P}} p$  and  $p \Vdash_{\text{HS}} \varphi$ , then  $q \Vdash_{\text{HS}} \varphi$ .*

**Lemma 2.1.24** (The Symmetry Lemma). *Let  $\pi$  be an automorphism of  $\mathbb{P}$ ,  $\dot{x}_0, \dots, \dot{x}_n \in \text{HS}$ , and  $\varphi$  be a formula, then*

$$p \Vdash_{\text{HS}} \varphi(\dot{x}_0, \dots, \dot{x}_n) \Leftrightarrow \pi p \Vdash_{\text{HS}} \varphi(\pi \dot{x}_0, \dots, \pi \dot{x}_n)$$

### 2.1.4 The c.t.m. Approach

## 2.2 Outline for the Informal Proof

We outline an informal proof of the relative consistency of  $\neg\text{AC}$  with ZF which we will formalize in the next chapter. In this proof, the relative consistency is proved by assuming the existence of a c.t.m. of ZF and constructing a model of  $\text{ZF} + \neg\text{AC}$  by forcing. This model is a symmetric extension called the basic Cohen model.

Let  $M$  be a c.t.m. of ZF,  $\mathbb{P}$  be the set of all finite partial functions from  $\omega \times \omega$  to  $2 = \{0, 1\}$  that are elements of  $M$ , and  $\leq_{\mathbb{P}}$  be  $\supseteq$ . Note that the maximum element  $1_{\mathbb{P}}$  is the empty set. Let  $\pi$  be a bijection on  $\omega$ .  $\pi$  induces an automorphism on  $\mathbb{P}$  defined as follows:

$$\begin{aligned} \text{dom}(\pi p) &= \{(\pi n, m) \mid (n, m) \in \text{dom}(p)\} \\ (\pi p)(\pi n, m) &= p(n, m) \end{aligned}$$

This automorphism further induces an automorphism on  $\mathbb{P}$ -names. Let  $\mathcal{G}$  be the group of all such automorphisms. For every finite  $e \subseteq \omega$ , let

$$\text{fix}(e) = \{\pi \in \mathcal{G} \mid \forall n \in e (\pi n = n)\}$$

Let  $\mathcal{F}$  be the set of all subgroups  $H$  of  $\mathcal{G}$  such that there exists a finite  $e \subseteq \omega$  with  $\text{fix}(e) \subseteq H$ . Note that  $\mathcal{F}$  is a normal filter on  $\mathcal{G}$ . Let  $\mathcal{N} = \text{HS}^{\mathcal{G}}$ . Since  $\mathcal{N}$  is a symmetric extension of  $M$ , it is a c.t.m. of ZF.

**Theorem 2.2.1.**  *$\mathcal{N}$  does not satisfy the well-ordering theorem.*

*Proof.* We outline the proof of this theorem as follows. For every  $n \in \omega$ , let  $a_n$  be the following real number (a subset of  $\omega$ ):

$$a_n = \{m \in \omega \mid \exists p \in G (p(n, m) = 1)\}$$

Since  $a_n$  are pairwise distinct,  $A = \{a_n \mid n \in \omega\}$  is an infinity set.  $A$  and every  $a_n$  are elements of  $\mathcal{N}$ .  $A$  serves as a counterexample to the well-ordering theorem in  $\mathcal{N}$ . Suppose for contradiction that  $A$  is well-ordered in  $\mathcal{N}$ , there exists an injection  $f$  from  $\omega$  to  $A$  in  $\mathcal{N}$ . Let  $\varphi(g, x, y)$  be a formula that represents the relation  $g(x) = y$ . For every  $n \in \omega$  with  $a_n \in \text{ran}(f)$ , there exists  $i \in \omega$  such that  $\mathcal{N} \models \varphi(f, i, a_n)$ . Thus there exists  $p \in G$  and hereditarily symmetric names  $\dot{f}, \dot{i}$  and  $\dot{a}_n$  for each of  $f, i, a_n$  such that

$$p \Vdash_{\text{HS}} \varphi(\dot{f}, \dot{i}, \dot{a}_n)$$

By choosing  $n$  and the names appropriately, we can find a bijection  $\pi$  on  $\omega$  such that the following conditions are additionally satisfied:

1.  $\pi \dot{f} = \dot{f}$
2.  $\pi \dot{i} = \dot{i}$
3.  $\pi n \neq n$
4. There exists a hereditarily symmetric name  $\dot{a}_{\pi n}$  of  $a_{\pi n}$  such that  $\pi \dot{a}_n = \dot{a}_{\pi n}$
5. There exists  $q \in G$  such that  $q \leq_{\mathbb{P}} p$  and  $q \leq_{\mathbb{P}} \pi p$

Note that some occurrence of  $\pi$  in above conditions refer to the induced automorphism on  $\mathbb{P}$  or  $\mathbb{P}$ -names.. By Lemma 2.1.24

$$\pi p \Vdash_{\text{HS}} \varphi(\pi \dot{f}, \pi \dot{i}, \pi \dot{a}_n)$$

Thus

$$\pi p \Vdash_{\text{HS}} \varphi(\dot{f}, \dot{i}, \dot{a}_{\pi n})$$

Therefore, by Lemma 2.1.23

$$q \Vdash_{\text{HS}} \varphi(\dot{f}, \dot{i}, \dot{a}_n) \text{ and } q \Vdash_{\text{HS}} \varphi(\dot{f}, \dot{i}, \dot{a}_{\pi n})$$

This means that  $\mathcal{N} \models \varphi(f, i, a_n)$  and  $\mathcal{N} \models \varphi(f, i, a_{\pi n})$ , which implies that  $f(i) = a_n$  and  $f(i) = a_{\pi n}$ . Since  $a_n$  and  $a_{\pi n}$  are distinct, this is a contradiction.  $\square$

## 2.3 Isabelle/ZF and Formalization in Prior Work

In this section, we introduce Isabelle/ZF, a proof assistant for ZF set theory, and the results from prior work used in the formalization of this study.

**The locale "forcing\_data" by Gunther et al.**

## Chapter 3

# Formalization of the Proof

In this chapter, we present the formalization of the relative consistency proof of  $\neg\text{AC}$  with ZF in Isabelle/ZF. We follow the outline presented in section 2.2. We choose *ZF-Constructible* by Paulson [11] as Isabelle theory <sup>1</sup>, which includes the ZF axioms along with useful definitions and lemmas built on top of them. We also reuse the formalization of forcing by Gunther et al. [1].

### 3.1 Defining Symmetric Extensions

First, we define symmetric extensions in Isabelle/ZF, formalizing automorphisms, normal filters, and hereditarily symmetric names.

#### $\mathbb{P}$ -names

We begin our work within the locale "forcing\_data" by Gunther et al. [1], in which a c.t.m. "M" and a notion of forcing "P" are fixed. The relation  $\leq_{\mathbb{P}}$  is denoted by "leq" and a maximum element  $1_{\mathbb{P}}$  is denoted by "one".

Gunther et al. did not provide an explicit formalization of the set of  $\mathbb{P}$ -names  $M^{\mathbb{P}}$  in their work. In this study, since we need to consider a subset of  $M^{\mathbb{P}}$ , we define  $M^{\mathbb{P}}$  explicitly. We define  $M_{\alpha}^{\mathbb{P}}$  as "P\_set( $\alpha$ )" and  $M^{\mathbb{P}}$  as "P\_names" as follows:

```
definition HP_set_succ :: "[i, i]  $\Rightarrow$  i" where
  "HP_set_succ(a, X)  $\equiv$  Pow(X  $\times$  P)  $\cap$  M"

definition P_set :: "i  $\Rightarrow$  i" where
  "P_set(a)  $\equiv$  transrec2(a, 0, HP_set_succ)"

definition P_names :: "i" where "P_names  $\equiv$  { x  $\in$  M .  $\exists$  a. Ord(a)  $\wedge$  x  $\in$  P_set(a) }"
```

Where "transrec2( $\alpha$ , x, H)" is the function defined in ZF library that returns x when  $\alpha$  is 0; when  $\alpha$  is a successor ordinal, it returns H( $\beta$ , transrec2( $\beta$ , x, H)) where  $\beta$  is the

---

<sup>1</sup>Note that this does not mean assuming  $V = L$ .

predecessor of  $\alpha$ ; and when  $\alpha$  is a limit ordinal, it returns  $\bigcup \beta \in \alpha. \text{transrec2}(\beta, x, H)$ . Note that in Isabelle/ZF, the notation " $\{x \in A. P(x)\}$ " denotes the set of elements in  $A$  that satisfy the predicate  $P$ .

## Automorphisms

To simplify the formalization, we declare and work within the locale "forcing\_data\_partial", where  $\mathbb{P}$  is assumed to be a partially ordered set. Note that "one" is the unique maximum element of  $\mathbb{P}$  in this locale.

```
locale forcing_data_partial = forcing_data +
  assumes leq_relation_on_P : "leq ∈ Pow(P × P)"
  and leq_partial_order : "partial_order_on(P, leq)"
```

We define the set of automorphisms on  $\mathbb{P}$ , denoted by "P\_auto", as follows:

```
definition is_P_auto :: "i ⇒ o" where
  "is_P_auto(π) ≡ π ∈ M ∧ π ∈ bij(P, P) ∧ (∀ p ∈ P. ∀ q ∈ P. p ≤ q ↔ π`p ≤ π`q)"

definition P_auto where "P_auto ≡ { π ∈ P → P. is_P_auto(π) }"
```

Where "bij(A, B)" is the set of bijections from  $A$  to  $B$  and "f`x" denotes the function application of  $f$  to  $x$ . We denote the induced automorphism on  $\mathbb{P}$ -names by  $\pi$  as " $\text{Pn\_auto}(\pi)$ ". Our definition of " $\text{Pn\_auto}$ " satisfies the equality given in Definition 2.1.14.

```
lemma Pn_auto :
  "x ∈ P_names ⇒ Pn_auto(π)`x = { <Pn_auto(π)`y, π`p> . <y, p> ∈ x }"
```

## Groups of Automorphisms and Normal Filters

First, we define the set of all subgroups of a group of automorphisms  $G$  that are in " $M$ ".

```
definition is_P_auto_group where
  "is_P_auto_group(G) ≡
    G ⊆ { π ∈ P → P. is_P_auto(π) }
    ∧ (∀ π ∈ G. ∀ τ ∈ G. π ∘ τ ∈ G)
    ∧ (∀ π ∈ G. converse(π) ∈ G)"

definition P_auto_subgroups where
  "P_auto_subgroups(G) ≡ { H ∈ Pow(G) ∩ M. is_P_auto_group(H) }"
```

Where " $\pi \circ \tau$ " denotes the composition of  $\pi$  and  $\tau$ . Then, we declare a locale "M\_symmetric\_system" that fixes a group of automorphisms " $\mathcal{G}$ " and a normal filter " $\mathcal{F}$ " on " $\mathcal{G}$ ".

```
locale M_symmetric_system = forcing_data_partial +
  fixes G F
  assumes G_in_M : "G ∈ M"
```

```

and  $\mathcal{G}_{P\_auto\_group}$  : "is_P_auto_group( $\mathcal{G}$ )"
and  $\mathcal{F}_{in\_M}$  : " $\mathcal{F} \in M$ "
and  $\mathcal{F}_{subset}$  : " $\mathcal{F} \subseteq P\_auto\_subgroups(\mathcal{G})$ "
and  $\mathcal{F}_{nonempty}$  : " $\mathcal{F} \neq 0$ "
and  $\mathcal{F}_{closed\_under\_intersection}$  : " $\forall A \in \mathcal{F}. \forall B \in \mathcal{F}. A \cap B \in \mathcal{F}$ "
and  $\mathcal{F}_{closed\_under\_supergroup}$  :
  " $\forall A \in \mathcal{F}. \forall B \in P\_auto\_subgroups(\mathcal{G}). A \subseteq B \rightarrow B \in \mathcal{F}$ "
and  $\mathcal{F}_{normal}$  : " $\forall H \in \mathcal{F}. \forall \pi \in \mathcal{G}. \{ \pi \circ \tau \circ converse(\pi). \tau \in H \} \in \mathcal{F}$ "

```

### Hereditarily Symmetric Names

We continue our work within the locale "M\_symmetric\_system". We define the set of hereditarily symmetric names "HS" in the same way as "P\_names". This approach makes it easier to use induction based on the  $\mathbb{P}$ -rank of names.

```

definition sym where "sym(x)  $\equiv$  {  $\pi \in \mathcal{G}. Pn\_auto(\pi) \`x = x$  }"

definition symmetric where "symmetric(x)  $\equiv$  sym(x)  $\in \mathcal{F}$ "

definition HHS_set_succ where
  "HHS_set_succ(a, X)  $\equiv$  {  $x \in P\_set(succ(a)). domain(x) \subseteq X \wedge symmetric(x)$  }"

definition HS_set where "HS_set(a)  $\equiv$  transrec2(a, 0, HHS_set_succ)"

definition HS where "HS  $\equiv$  {  $x \in P\_names. \exists a. Ord(a) \wedge x \in HS\_set(a)$  }"

```

The following lemma shows our definition of "HS" is equivalent to Definition 2.1.17.

```

lemma HS_iff: " $x \in HS \leftrightarrow x \in P\_names \wedge domain(x) \subseteq HS \wedge symmetric(x)$ "

```

### Symmetric Extensions

Finally, we define symmetric extensions.

```

definition SymExt where "SymExt(G)  $\equiv$  { val(G, x).  $x \in HS$  }"

```

Where "val(G, x)" is the interpretation  $x^G$  formalized by Gunther et al. [1].

We declare a locale "M\_symmetric\_system\_G\_generic", which combines the locale "M\_symmetric\_system" and the locale "G\_generic". "G\_generic" is a locale, defined by Gunther et al. [1], fixes a generic filter "G" on "P".

```

locale M_symmetric_system_G_generic = M_symmetric_system + G_generic

```

The following lemmas can be proved relatively easily in this locale.

```

lemma M_subset_SymExt : " $M \subseteq \text{SymExt}(G)$ "
lemma SymExt_subset_GenExt : " $\text{SymExt}(G) \subseteq \text{GenExt}(G)$ "
lemma Transset_SymExt : " $\text{Transset}(\text{SymExt}(G))$ "
lemma SymExt_countable : " $\text{nat} \approx \text{SymExt}(G)$ "

```

Where " $\text{GenExt}(G)$ " is the generic extension generated by " $G$ ", " $\text{Transset}(A)$ " denotes that " $A$ " is transitive, and " $\text{nat} \approx A$ " denotes that " $A$ " has the same cardinality as the set of natural numbers.

### Formulas Representing These Definitions

In our proofs, we need to use the separation and replacement axioms in  $M$  based on formulas that include the concepts defined in this section. Therefore, it is necessary to represent these concepts as internalized formulas.

We utilized the automated formula synthesis method by Gunther et al. [1]; however, it could not applied to complex formulas, and re-verifying the properties of the derived formulas in an accessible form remained a repetitive task.

Example of the formulas we defined includes " $\text{is\_P\_auto\_group}$ ", " $\text{is\_P\_name\_fm}$ ", " $\text{is\_Pn\_auto\_fm}$ ", and " $\text{is\_HS\_fm}$ ", which represent " $P\_names$ ", " $Pn\_auto$ " and " $HS$ " respectively. The correctness of these formulas is verified by proving the following lemmas.

```

lemma sats_is_P_name_fm_iff :
  fixes env i j x
  assumes "env ∈ list(M)" "i < length(env)" "j < length(env)"
    "nth(i, env) = P" "nth(j, env) = x"
  shows "sats(M, is_P_name_fm(i, j), env) ↔ x ∈ P_names"

lemma sats_is_Pn_auto_fm_iff :
  fixes x n v env i j k l
  assumes "i < length(env)" "j < length(env)" "k < length(env)" "l < length(env)"
    "nth(i, env) = P" "nth(j, env) = n" "nth(k, env) = x" "nth(l, env) = v"
    "env ∈ list(M)" "n ∈ P_auto"
  shows "sats(M, is_Pn_auto_fm(i, j, k, l), env) ↔ x ∈ P_names ∧ v = Pn_auto(n) ` x"

lemma sats_is_HS_fm_iff :
  fixes x i j env
  assumes "env ∈ list(M)" "i < length(env)" "j < length(env)"
    "nth(i, env) = <F, G, P, P_auto>" "nth(j, env) = x"
  shows "sats(M, is_HS_fm(i, j), env) ↔ x ∈ HS"

```

Note that these formulas take as parameters various elements fixed within the locale, such as the notion of forcing " $P$ ", the normal filter " $F$ ", etc.

## 3.2 Proving Symmetric Extensions are Models of ZF

In this section, we prove that symmetric extensions are models of ZF. Initially, we attempted to prove this based on its proof in Karagila's lecture note [8], but this approach was unsuccessful. Instead, we prove this by using the relativized forcing relation  $\Vdash_{\text{HS}}$ . Karagila's approach is applicable to a more general situation, without using forcing or generic and symmetric extensions, whereas our approach is more specific to the context of symmetric extensions.

### Unsuccessful Approach

In Karagila's lecture note [8], the fact that symmetric extensions are models of ZF is proven by showing that they are internal models of the corresponding generic extension generated by the same notion of forcing and generic filter, using the following proposition:

**Proposition 3.2.1.** *If  $\mathcal{N}$  is a transitive class that is almost universal and satisfies  $\Delta_0$ -separation, then  $\mathcal{N}$  is a inner model of ZF.*

Where a class  $\mathcal{N}$  is said to be *almost universal*<sup>2</sup> if every subset  $X \subseteq \mathcal{N}$  is included in some  $Y \in \mathcal{N}$ . Furthermore,  $\mathcal{N}$  is said to satisfy  $\Delta_0$ -separation if it satisfies the separation axioms for formulas with only bounded quantifiers.

We could prove that the symmetric extension " $\text{SymExt}(G)$ " is transitive, almost universal and satisfies  $\Delta_0$ -separation. However, we could not prove that " $\text{SymExt}(G)$ " is actually a class in " $\text{GenExt}(G)$ ". To do this, we needed to explicitly construct a formula that represents the set of all names " $P_{\text{names}}$ " on the ground model " $M$ " in " $\text{GenExt}(G)$ ". However, we were unable to construct such a formula.

### The Relativized Forcing Relation $\Vdash_{\text{HS}}$

Instead, we formalize the relativized forcing relation  $\Vdash_{\text{HS}}$  and prove that symmetric extensions are models of ZF using it. This relation is defined by replacing all quantifiers restricted to  $M^P$  in the definition of the forcing relation with quantifiers restricted to HS. By modifying the formalization of forcing by Gunther et al. [1]<sup>3</sup>, we were able to define this relation and prove its properties relatively easily.

Within the locale " $M_{\text{symmetric\_system}}$ ", we define this relation in the form " $p \Vdash_{\text{HS}} \varphi \text{ env}$ ", which has the same form as the formalization of the original forcing relation " $p \Vdash \varphi \text{ env}$ " by Gunther et al. Our definition satisfies the required properties, such as Theorem 2.1.21, Corollary 2.1.22, and Lemma 2.1.23.

```
lemma HS_truth_lemma:
  assumes
    " $\varphi \in \text{formula}$ " " $M_{\text{generic}}(G)$ "
```

<sup>2</sup>Jech [5], Theorem 13.9.

<sup>3</sup>For non-atomic formulas, the definition of the forcing relation by Gunther et al. [1] actually differs from Definition 2.1.11. In their definition, the unique logical connective is NAND, and the unique quantifier is  $\forall$ . Additionally, in the definition, quantifiers restricted to  $M$  appear instead of those restricted to  $M^P$ .

```

shows
  " $\bigwedge \text{env. env} \in \text{list}(\text{HS}) \Rightarrow \text{arity}(\varphi) \leq \text{length}(\text{env}) \Rightarrow$ "
  " $(\exists p \in G. p \Vdash_{\text{HS}} \varphi \text{ env}) \leftrightarrow \text{SymExt}(G), \text{map}(\text{val}(G), \text{env}) \models \varphi$ "

lemma definition_of_forcing_HS:
  assumes
    " $p \in P$ " " $\varphi \in \text{formula}$ " " $\text{env} \in \text{list}(\text{HS})$ " " $\text{arity}(\varphi) \leq \text{length}(\text{env})$ "
  shows
    " $(p \Vdash_{\text{HS}} \varphi \text{ env}) \leftrightarrow$ "
    " $(\forall G. M_{\text{generic}}(G) \wedge p \in G \rightarrow \text{SymExt}(G), \text{map}(\text{val}(G), \text{env}) \models \varphi)$ "

lemma HS_strengthening_lemma:
  assumes
    " $p \in P$ " " $\varphi \in \text{formula}$ " " $r \in P$ " " $r \leq p$ "
  shows
    " $\bigwedge \text{env. env} \in \text{list}(M) \Rightarrow \text{arity}(\varphi) \leq \text{length}(\text{env}) \Rightarrow p \Vdash_{\text{HS}} \varphi \text{ env} \Rightarrow r \Vdash_{\text{HS}} \varphi \text{ env}$ "

```

## The Symmetry Lemma

Additionally, we prove the symmetry lemma (Lemma 2.1.24). Similar to the definitions of the forcing relations, this lemma is proven using a complex induction. First, we prove this lemma for atomic formulas in the forcing language, and then extend the proof to other formulas by structural induction on the formula. For atomic formulas  $\dot{x} = \dot{y}$  and  $\dot{x} \in \dot{y}$ , the definition is mutually recursive, so the lemma for these formulas is proven simultaneously by induction on the  $\mathbb{P}$ -rank of  $\dot{x}$  and  $\dot{y}$ . Specifically, we show these induction steps:

```

have MEM_step: " $\bigwedge a. \text{Ord}(a) \Rightarrow \forall b \in a. Q(\text{EQ}, b, b) \Rightarrow \forall b \in a. Q(\text{MEM}, b, a)$ "
have EQ_step : " $\bigwedge a. \text{Ord}(a) \Rightarrow \forall b \in a. Q(\text{MEM}, b, a) \Rightarrow Q(\text{EQ}, a, a)$ "

```

Where " $Q(\text{MEM}, \alpha, \beta)$ " and " $Q(\text{EQ}, \alpha, \beta)$ " are predicates representing that the symmetry lemma holds for  $\dot{x} \in \dot{y}$  and  $\dot{x} = \dot{y}$ , respectively, for any  $\dot{x}$  with  $\mathbb{P}$ -rank at most  $\alpha$  and  $\dot{y}$  with  $\mathbb{P}$ -rank at most  $\beta$ .

The proof for non-atomic formulas is relatively straightforward, and finally, this completes the proof of the lemma.

```

lemma symmetry_lemma:
  fixes  $\varphi \pi$ 
  assumes " $\varphi \in \text{formula}$ " " $\text{is\_P\_auto}(\pi)$ " " $\pi \in \mathcal{C}$ "
  shows " $\bigwedge \text{env } p. \text{env} \in \text{list}(\text{HS}) \Rightarrow \text{arity}(\varphi) \leq \text{length}(\text{env}) \Rightarrow p \in P \Rightarrow$ "
  " $p \Vdash_{\text{HS}} \varphi \text{ env} \leftrightarrow \pi ` p \Vdash_{\text{HS}} \varphi \text{ map}(\lambda x. Pn\_auto(\pi) ` x, \text{env})$ "

```

## Separation

Using the symmetry lemma, we prove the axiom schema of separation for symmetric extensions within the locale " $M_{\text{symmetric\_system\_G\_generic}}$ ". Our proof is based on



Karagila's proof [8] of  $\Delta_0$ -separation for symmetric extensions. However, since we use the  $\Vdash_{\text{HS}}$  relation instead of  $\Vdash$ , we can prove separation for any formula, not just  $\Delta_0$  formulas.

The separation for the formula  $\varphi$  means that for any elements  $x, p$  in the symmetric extension, the set  $X = \{y \in x \mid \varphi(y, p)\}$  is also an element of it. To prove this, we show that  $\dot{X} = \{\langle y, p \rangle \in \text{dom}(x) \times \mathbb{P} \mid p \Vdash_{\text{HS}} \varphi(\dot{y}, \dot{p})\}$  is a hereditarily symmetric name of  $X$ , i.e.,  $\dot{X} \in \text{HS}$  and  $\dot{X}^G = X$ .

```
lemma sep_forces_pair_in_HS :
  fixes x env φ
  assumes "x ∈ HS" "env ∈ list(HS)" "φ ∈ formula" "arity(φ) ≤ succ(length(env))"
  shows "{ <y, p> ∈ domain(x) × P. p ⊩_HS φ [y] @ env } ∈ HS"

lemma SymExt_separation :
  fixes x env φ
  assumes "x ∈ SymExt(G)" "env ∈ list(SymExt(G))"
    "φ ∈ formula" "arity(φ) ≤ succ(length(env))"
  shows "{ y ∈ x. sats(SymExt(G), φ, [y] @ env) } ∈ SymExt(G)"
```

## Replacement

Now that we have proven separation, to prove the axiom schema of replacement for symmetric extensions, it is sufficient to show that for any formula  $\varphi$  and any elements  $x, p$  in a symmetric extension  $\mathcal{N}$ , we can find a set  $S \in \mathcal{N}$  such that :

$$\mathcal{N} \models \exists z \varphi(y, z, p) \text{ iff } \mathcal{N} \models \exists z (z \in S \wedge \varphi(y, z, p)) \text{ for all } y \in x$$

This is because such a set  $S$  contains the image of  $x$  under the class function defined by  $\varphi^4$ , which is the set that would be obtained by applying the replacement axiom. To prove this, we show the following lemma, which rewrites the above statement in terms of the forcing relation  $\Vdash_{\text{HS}}$ , and then prove it.

```
lemma ex_hs_subset_contains_witnesses :
  fixes φ env x
  assumes "φ ∈ formula" "env ∈ list(M)" "arity(φ) ≤ 2 #+ length(env)" "x ∈ M"
  shows "∃S. S ∈ M ∧ S ⊆ HS ∧ (∀p ∈ G. ∀y ∈ domain(x).
    (∃z ∈ HS. p ⊩_HS φ ([y, z] @ env)) ↔ (∃z ∈ S. p ⊩_HS φ ([y, z] @ env)))"

lemma ex_SymExt_elem_contains_witnesses :
  fixes φ env x
  assumes "φ ∈ formula" "env ∈ list(SymExt(G))"
    "arity(φ) ≤ 2 #+ length(env)" "x ∈ SymExt(G)"
  shows "∃S ∈ SymExt(G). ∀y ∈ x.
```

---

<sup>4</sup>If  $\varphi$  defines a class function.

```

(( $\exists z \in \text{SymExt}(G).$  sats(SymExt(G),  $\varphi$ , [y, z] @ env))
 $\leftrightarrow$  ( $\exists z \in S.$  sats(SymExt(G),  $\varphi$ , [y, z] @ env)))"

```

Additionally, the following lemma was helpful in these proofs. This is also frequently used in the proofs of the axioms.

```

lemma ex_separation_base :
  fixes X
  assumes "X  $\subseteq$  HS" "X  $\in$  M"
  shows " $\exists S \in \text{SymExt}(G).$  { val(G, x). x  $\in$  X }  $\subseteq$  S"

```

This lemma states that for any set of hereditarily symmetric names  $X \in M$ , there exists a set  $S$  in the symmetric extension such that  $X^G \subseteq S$ .

Then, we prove the axiom schema of replacement for symmetric extensions.

```

lemma SymExt_replacement :
  fixes  $\varphi$  env
  assumes " $\varphi \in \text{formula}$ " "arity( $\varphi$ )  $\leq$  2 #+ length(env)" "env  $\in$  list(SymExt(G))"
  shows
    "strong_replacement(##SymExt(G),  $\lambda x y.$  sats(SymExt(G),  $\varphi$ , [x, y] @ env))"

```

## Other Axioms

Other axioms of ZF can be proven relatively easily in symmetric extensions by using separation, replacement and the lemma "ex\_separation\_base". For example, the axiom of pairing can be proven as follows. Let  $a, b$  be elements of a symmetric extension  $\mathcal{N}$ . Then, we can take hereditarily symmetric names  $\dot{a}, \dot{b}$  of  $a, b$ , respectively. By the lemma "ex\_separation\_base", we can find a set  $S$  in  $\mathcal{N}$  such that  $\{\dot{a}, \dot{b}\}^G \subseteq S$ . In other words,  $\{a, b\} \subseteq S$ . Then by the separation axiom,  $\{x \in S \mid x = a \vee x = b\} = \{a, b\}$  is an element of  $\mathcal{N}$ .

Finally we obtain the following lemma, which states that symmetric extensions are models of ZF.

```

theorem SymExt_sats_ZF : "SymExt(G)  $\models$  ZF"

```

Furthermore, since symmetric extensions are transitive, it can be shown that they are instances of the locale "M\_ZF\_trans". As a result, many results concerning ZF set theory formalized in this locale can now be applied to symmetric extensions.

```

lemma SymExt_M_ZF_trans : "M_ZF_trans(SymExt(G))"

```

## 3.3 Constructing a Model of ZF+ $\neg$ AC

In this section, we define the basic Cohen model which is a symmetric extension that does not satisfy the axiom of choice, and prove that it is a model of ZF+ $\neg$ AC, as outlined in section 2.2. First, we work within the locale "M\_ctm" which fixes a c.t.m. "M" of ZF.

## The Notion of Forcing

the notion of forcing for the basic Cohen model is defined as the set of finite partial functions from  $\omega \times \omega$  to 2 that are elements of the ground model, with the order relation given by the superset relation. We define these as "Fn" and "Fn\_leq" in Isabelle/ZF.

```
definition Fn where "Fn  $\equiv$  { f  $\in$  Pow((nat  $\times$  nat)  $\times$  2)  $\cap$  M.  

  function(f)  $\wedge$  domain(f)  $\subseteq$  nat  $\times$  nat  $\wedge$  finite_M(domain(f))  $\wedge$  range(f)  $\subseteq$  2 }"  

  

definition Fn_leq where "Fn_leq  $\equiv$  { <f, g>  $\in$  Fn  $\times$  Fn. g  $\subseteq$  f }"
```

Where "nat" denotes  $\omega$  and "finite\_M(A)" denotes that the set "A" is finite. From the following lemma, this ordered set is an instance of the locale "forcing\_data\_partial", meaning that this is indeed a notion of forcing and forms a partially ordered set with the empty set as the maximum element.

```
lemma Fn_forcing_data_partial : "forcing_data_partial(Fn, Fn_leq, 0, M, enum)"
```

Where "enum" is the enumeration function of the set "M" fixed in the locale "M\_ctm".

## The Group of Automorphisms and the Normal Filter

Next, we define the group of automorphisms "Fn\_perms" on the forcing notion "Fn".

```
definition nat_perms where "nat_perms  $\equiv$  bij(nat, nat)  $\cap$  M"  

definition Fn_perm where "Fn_perm(f, p)  $\equiv$  { <<f`n, m>, l> . <<n, m>, l>  $\in$  p }"  

definition Fn_perm' where "Fn_perm'(f)  $\equiv$  { <p, Fn_perm(f, p)> . p  $\in$  Fn }"  

definition Fn_perms where "Fn_perms  $\equiv$  { Fn_perm'(f). f  $\in$  nat_perms }"
```

Note that "Fn\_perm'(f)" is the automorphism induced by the bijection "f" on  $\omega$ , and "Fn\_perm(f, p)" is the value of the automorphism "Fn\_perm'(f)" on the partial function "p". We prove that "Fn\_perm'(f)" is indeed an automorphism on "Fn" and "Fn\_perms" is a group of automorphisms on "Fn", i.e., closed under composition and the inverse operation, and contains the identity function.

```
lemma Fn_perm'_is_P_auto :  

  fixes f  

  assumes "f  $\in$  nat_perms"  

  shows "forcing_data_partial.is_P_auto(Fn, Fn_leq, M, Fn_perm'(f))"  

  

lemma Fn_perms_group :  

  "forcing_data_partial.is_P_auto_group(Fn, Fn_leq, M, Fn_perms)"
```

Then, we define the normal filter "Fn\_perms\_filter" on the group of automorphisms "Fn\_perms", and verify that our definition satisfies the prerequisites of the locale "M\_symmetric\_system". This completes both the definition of the basic Cohen model and the proof that it is a model of ZF.

```

definition Fn_perms_filter where "Fn_perms_filter  $\equiv$  {
  H  $\in$  forcing_data_partial.P_auto_subgroups(Fn, Fn_leq, M, Fn_perms).
   $\exists E \in \text{Pow}(\text{nat}) \cap M. \text{finite\_M}(E) \wedge \text{Fix}(E) \subseteq H$ 
}"

```

```

lemma Fn_M_symmetric_system :
  "M_symmetric_system(Fn, Fn_leq, 0, M, enum, Fn_perms, Fn_perms_filter)"

```

Finally, we prove that the basic Cohen model does not satisfy the axiom of choice. To do this, we construct a set  $A$  that cannot be well-ordered in it. We work within the locale "M\_symmetric\_system" with parameters, as defined above, fixed to generate the basic Cohen model.

We formalize  $a_n$  and  $A$ , introduced in section 2.2, as "binmap\_row( $G$ ,  $n$ )" and "binmap( $G$ )", respectively, for given generic filter " $G$ ".

```

definition binmap_row where
  "binmap_row(G, n)  $\equiv$  { m  $\in$  nat.  $\exists p \in G. p \restriction n, m = 1$  }"

definition binmap where "binmap(G)  $\equiv$  { binmap_row(G, n). n  $\in$  nat }"

```

Then, we prove that  $a_n$  are pairwise disjoint, which implies that  $A$  is an infinite set.

```

lemma binmap_row_distinct :
  fixes G n m
  assumes "M_generic(G)" "n  $\in$  nat" "m  $\in$  nat" "n  $\neq$  m"
  shows "binmap_row(G, n)  $\neq$  binmap_row(G, m)"

```

Using this lemma, we prove that  $p \Vdash_{\text{HS}} \dot{F}$  is a injection from  $\omega$  to  $\dot{A}$  leads a contradiction for any  $p \in \text{Fn}$  and  $\dot{F} \in \text{HS}$ . This means that there is no injection from  $\omega$  to  $A$  in the basic Cohen model.

```

lemma no_injection :
  fixes F' p0
  assumes "F'  $\in$  HS" "p0  $\in$  Fn"
  "ForcesHS(p0, injection_fm(0, 1, 2), [check(nat), binmap', F'])"
  shows False

```

Where "ForcesHS( $p$ ,  $\varphi$ ,  $\text{env}$ )" is another notation for " $p \Vdash_{\text{HS}} \varphi \text{ env}$ ", and "check(nat)" and "binmap'" are the hereditarily symmetric names of "nat" and "binmap", respectively.

Finally, from this lemma, we can show that assuming  $A$  is well-orderable leads to a contradiction, because  $A$  is an infinite set, which implies the existence of an injection from  $\omega$  to  $A$ .

```

lemma no_wellorder :
  fixes r G
  assumes
    "M_generic(G)" "wellordered(##SymExt(G), binmap(G), r)" "r  $\in$  SymExt(G)"

```

| shows False

From the above, we complete the proof that the basic Cohen model is a model of  $\text{ZF} + \neg \text{AC}$ . Our result is stated in the following theorem, with no additional assumptions from any locale.

## **Chapter 4**

# **Conclusion and Future Work**

### **Acknowledgements**

# Bibliography

- [1] Emmanuel Gunther, Miguel Pagano, and Pedro Sánchez Terraf. “Formalization of Forcing in Isabelle/ZF”. In: *Archive of Formal Proofs* (May 2020). <https://isa-afp.org/entries/Forcing.html>, Formal proof development. ISSN: 2150-914x.
- [2] Emmanuel Gunther et al. “The Independence of the Continuum Hypothesis in Isabelle/ZF”. In: *Archive of Formal Proofs* (Mar. 2022). [https://isa-afp.org/entries/Independence\\_CH.html](https://isa-afp.org/entries/Independence_CH.html), Formal proof development. ISSN: 2150-914x.
- [3] Jesse Michael Han and Floris van Doorn. “A formal proof of the independence of the continuum hypothesis”. In: *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs, CPP 2020, New Orleans, LA, USA, January 20-21, 2020*. 2020.
- [4] M. Randall Holmes and Sky Wilshaw. *NF is Consistent*. 2024. eprint: arXiv:1503.01406.
- [5] Thomas J Jech. *Set Theory*. en. 3rd ed. Springer Monographs in Mathematics. Berlin, Germany: Springer, Oct. 2002.
- [6] Thomas J Jech. *The axiom of choice*. Dover Books on Mathematics. Mineola, NY: Dover Publications, July 2008.
- [7] Kenneth Kunen. *Set Theory*. London, England: College Publications, Nov. 2011.
- [8] *Lecture Notes: Forcing & Symmetric Extensions*. Accessed: 2024-10-29. 2023. URL: <https://karagila.org/files/Forcing-2023.pdf>.
- [9] Lawrence C. Paulson. “A fixedpoint approach to implementing (Co)inductive definitions”. In: *Automated Deduction — CADE-12*. Ed. by Alan Bundy. Berlin, Heidelberg: Springer Berlin Heidelberg, 1994, pp. 148–161. ISBN: 978-3-540-48467-7.
- [10] Lawrence C. Paulson. “The Reflection Theorem: A Study in Meta-theoretic Reasoning”. In: *Automated Deduction—CADE-18*. Springer Berlin Heidelberg, 2002, pp. 377–391. ISBN: 9783540456209. DOI: 10.1007/3-540-45620-1\_31. URL: [http://dx.doi.org/10.1007/3-540-45620-1\\_31](http://dx.doi.org/10.1007/3-540-45620-1_31).
- [11] Lawrence C. Paulson. “The Relative Consistency of the Axiom of Choice — Mechanized Using Isabelle/ZF”. In: *Logic and Theory of Algorithms*. Springer Berlin Heidelberg, pp. 486–490. ISBN: 9783540694076. DOI: 10.1007/978-3-540-69407-6\_52. URL: [http://dx.doi.org/10.1007/978-3-540-69407-6\\_52](http://dx.doi.org/10.1007/978-3-540-69407-6_52).

- [12] LawrenceC. Paulson and Krzysztof Grabczewski. “Mechanizing set theory: Cardinal arithmetic and the Axiom of Choice”. In: *Journal of Automated Reasoning* 17.3 (Dec. 1996). issn: 1573-0670. DOI: 10.1007/bf00283132. URL: <http://dx.doi.org/10.1007/BF00283132>.
- [13] Gordon Plotkin, Colin P. Stirling, and Mads Tofte. “A Fixedpoint Approach to (Co)Inductive and (Co)Datatype Definitions”. In: *Proof, Language, and Interaction: Essays in Honour of Robin Milner*. 2000, pp. 187–211.